MACRO 

H15 X  

#___________________________________________________________ 

# AMC Software.  Minitab local macro

# Program to calculate Huber's H15 robust mean and standard deviation.   

# Last modified:  05/02/01

# Input argument is a single column X, which is not changed.

# Executed in a Minitab session by using the command %H15 C

# where C is a column, assuming that the program is 

# saved as a text file in the Minitab subdirectory <MACROS> under the 

# filename <H15.MAC>.

# If the macro file is saved elsewhere, it must be called by using the session

# command %<pathname><filename>.

# The program will run under versions of Minitab from 10Xtra onwards.

# The robust technique is designed for medium sized data sets: 

# Results from small sets should be treated with caution.

# Misleading results may be obtained if the program is used on data sets

# that are multimodal or strongly skewed or almost invariant.

# The algorithm is based on Analyst, 1989, 114, 1693. 

# See also AMC Technical Brief no 6 on www.rsc.org/lap/rsccom/amc/amc_index.htm

#__________________________________________________________ 

#i/o arguments 

MCOLUMN X 

#column of input data 

#other declarations 

MCONSTANTS M S 
#robust mean and sd 

MCOLUMNS DIFF Y
#scratch columns 

MCONSTANT SA 

#scratch constant 

MCONSTANT MAD 
#median absolute difference 

MCONSTANT TOL 

#terminates iteration if < 0.00001 

MCONSTANT HIGH LOW  #limits for winsorisation 

MCONSTANT KOUNT 
#loop counter 

MCONSTANT MAX MIN 
#highest and lowest data values

MCONSTANT COLUMN
#column name

MCONSTANT ROBMEAN 
#output name

MCONSTANT ROBSD
#ditto

#_____________________________________________________________

# set up starting values  

IF N(X)< 6


NOTE DATA SET TOO SMALL






LET M = '*' 


LET S = '*' 


EXIT 

ELSE




 


LET M = MEDIAN(X) 


LET DIFF = X - M 


LET MAD= MEDIAN(ABSOLUTE(DIFF)) 


LET S = MAD/0.6745 

ENDIF 

# initialise variables for iteration

LET TOL = 0.1 

LET KOUNT=0 

LET MAX=MAXIMUM(X)*1.1 

LET MIN=MINIMUM(X)*0.9 

# iteration 

WHILE (TOL > 0.00001 AND KOUNT<100) OR KOUNT<10



COPY X Y  


LET SA = S 


LET LOW = M-1.5*S 


LET HIGH = M + 1.5*S 


#winsorise raw data and update estimates


CODE (MIN:LOW) LOW  (HIGH:MAX) HIGH Y Y 


LET M = MEAN(Y) 


LET S = STDEV(Y)/0.882 


#check for sufficient convergence


IF SA > 0 



LET TOL = ABSOLUTE(SA-S)/SA 


ELSE 



NOTE PROBLEM WITH NEAR ZERO SD - TERMINATED



LET M = '*' 



LET S = '*' 



EXIT 


ENDIF 


LET KOUNT=KOUNT+1 

ENDWHILE 

#output

NOTE

NOTE ROBUST MEAN AND STANDARD DEVIATION

NOTE

NOTITLE

LET ROBMEAN = M

LET ROBSD = S

KKNAME COLUMN X

PRINT COLUMN ROBMEAN ROBSD 

NOTE

TITLE 

IF  KOUNT > 99 


NOTE WARNING - ITERATION NOT CONVERGED 

ENDIF 

ENDMACRO 

#____________________________________________________________ 

